
Resumen 

Desde que Pigou (1920) propuso que la internalización de las 
externalidades negativas podría hacerse a través de un impuesto que 
corrigiera la diferencia entre costos privados y costos sociales, de forma tal 
que se recuperaran las condiciones de eficiencia económica, prácticamente 
en todas las legislaciones hidráulicas del mundo, se ha aplicado esta 
recomendación, sin embargo, en condiciones de gran escasez de agua, 
podría pensarse que gravar la cantidad de agua incorporada definitivamente 
al producto final, el uso consuntivo, podría ser más adecuado que gravar la 
producción de agua residual, logrando de esta manera, además de 
recuperar las condiciones de eficiencia social, un doble dividendo: incentivar 
el reciclaje de agua.  

En este trabajo se evalúa el efecto que el cobro de un impuesto al uso 
consuntivo y a la producción de agua residual tendría sobre una economía 
simplificada. Los resultados obtenidos se contrastan utilizando información 
de dos economías con muy diferentes dotaciones de agua. La comparación 
se hace utilizando un modelo de equilibrio general computable del tipo 
Arrow-Debreu, introduciendo un gobierno que recauda los impuestos 
propuestos en una economía previamente distorsionada y distribuye la 
recaudación vía una transferencia directa al ingreso de los consumidores. El 
modelo se calibra y se resuelve siguiendo la metodología propuesta por 
Shoven y Whalley (1984). 

Abstract 

Since Pigou (1920) proposed that the internalization of the negative 
externalities could have done calculating a tax in order to correct the 
difference between private costs and social costs; this will recover the 
conditions of economic efficiency in all the hydraulic legislations of the 
world, however, this advice has been applied in conditions of great 
shortage; it could be thought that to burden the consumptive use could be 
more suitable than to burden the production of residual water reaching this 
way not only the recovering of social efficiency but also a double dividend: 
to stimulate the water recycling. 

This work evaluates the effect of the tax charging to the consumptive 
use will have over a fake economy. The achieved results of a residual water 
production tax are compared with the use consumptive tax. The contrast is 
done using a model of computable general equilibrium with the Arrow-
Debreu assumptions relaxed with the introduction of a government that 
collect the proposed taxes in a previously distorted economy. The collection 
distributes through a direct transference to the revenue of the consumers, 



 

 

the model calibrates and resolves following the methodology proposed by 
Shoven & Whalley (1984). 

Introducción* 

En México, la contaminación de los cuerpos de agua y la escasez relativa de 
agua útil son problemas que la Comisión Nacional del Agua (CNA) intenta 
resolver aplicando políticas públicas cuyo objetivo principal es incentivar el 
reciclamiento del agua y la disminución de la emisión de aguas residuales. En 
este trabajo se comparan dos instrumentos de política, uno utilizado 
actualmente por la CNA: el cobro de un impuesto a la emisión de aguas 
residuales y el otro, un impuesto al uso consuntivo. La comparación de ambos 
instrumentos de política se hace utilizando un modelo de equilibrio general 
computable calibrado para datos de las cuencas hidrográficas Grijalva-
Usumacinta y Nazas-Aguanaval.  

El trabajo está organizado de la siguiente manera: en una primera parte se 
presentan los antecedentes del tema; a continuación se describe el modelo que 
se utilizará para la evaluación, se comenta la forma de calibración y solución 
del mismo y, finalmente, se presentan los resultados y las conclusiones a las 
que da lugar la aplicación del modelo. 

 

Antecedentes 

Desde el punto de vista económico, el problema de la contaminación del agua 
puede ser visto como la ausencia de precios para el servicio medioambiental 
del agua, por tanto para corregir estas fallas, habría que introducir algún 
“precio” que proporcione las señales adecuadas para economizar el uso del 
bien escaso, Kneese y Bower(1968). 

Los agentes económicos necesitan ser confrontados con un “precio” igual al 
costo marginal de la externalidad de sus actividades contaminadoras para 
inducirlos a internalizar el costo de sus actividades económicas. Una opción es 
que el “precio” tome la forma de un “impuesto pigouviano”, es decir, un 
impuesto que se aplica al agente contaminador y que sea igual al daño social 
marginal causado por el agente. 

Otra opción es que se establezca un subsidio unitario por la reducción de 
contaminación y los permisos de emisiones intercambiables en el mercado.  

La política de los subsidios unitarios, a pesar que genera los mismos costos 
de oportunidad que el impuesto pigouviano, tiene efectos muy distintos sobre 
los beneficios de las empresas y por ende, sobre la cantidad de empresas y del 
nivel de contaminación en el largo plazo. Probablemente esta sea la causa por 

la cual los subsidios no sean tan utilizados en la práctica como los impuestos 
pigouvianos.  

En contraste, en un mundo de previsión perfecta, la autoridad podría optar 
con total simetría entre fijar un impuesto igual al tamaño del costo causado y 
así obtener la cantidad eficiente de contaminantes o colocar permisos de 
emisión en una cantidad igual a la cantidad óptima y permitir que las empresas 
contaminadoras pujen por ellos.  

La solución a través del mercado satisface las condiciones tanto de 
eficiencia en la contaminación en el corto plazo, como de las decisiones de 
entrada y salida de la industria en el largo plazo. La autoridad puede fijar tanto 
la cantidad como el precio y llegar al mismo resultado. Sin embargo, estos 
resultados dependen del supuesto de previsión perfecta, ya que con 
información incompleta los resultados podrían variar de manera significativa. 
Weitzman (1974), estableció las condiciones bajo las cuales las ganancias 
esperadas en bienestar bajo un impuesto unitario exceden, igualan o están por 
debajo que las obtenidas  en un sistema de permisos comerciable también 
conocidos como cuotas.  

A pesar de la sencillez de la solución encontrada a través de los impuestos 
pigouvianos, esta línea de trabajo ha sido muy atacada por los seguidores de 
Coase (1960). El argumento de éstos es que, en ausencia de costos de 
transacción y de comportamiento estratégico, las distorsiones asociadas con las 
externalidades serán resueltas con negociaciones voluntarias entre las partes 
interesadas, por lo que sería innecesario inducir a los agentes económicos a 
través de un impuesto. De hecho, en un marco como el que Coase propone, la 
introducción de un impuesto pigouviano sería por sí misma una distorsión 
indeseable, Turvey (1963).  

En México es el Estado, vía la CNA, quien administra los recursos hidráulicos 
del país, para lo cual tiene la opción de instrumentar las políticas públicas que 
considere adecuadas. La CNA ha optado por aplicar el cobro de un impuesto 
pigouviano denominado “Derechos por uso o aprovechamiento de bienes del 
dominio público de la nación como cuerpos receptores de descargas de aguas 
residuales”. La efectividad de este impuesto ya ha sido cuestionada, Bravo y 
Ortiz (2001). El cobro de este derecho no ha sido exitoso, ni desde el aspecto 
recaudatorio, ni desde el punto de vista de mejora de la calidad del agua.  

El cobro de este impuesto se ha modificado varias veces desde su 
implementación en 1992, se han cambiado desde los plazos para construir la 
infraestructura de tratamiento, lo que garantiza postergar el pago del impuesto 
hasta el plazo fijado ex profeso por la CNA,  hasta los parámetros significativos 
para el cálculo del cobro. Probablemente esta falta de consistencia sea una de 
las principales causas del incumplimiento de los objetivos con el que fue 
concebido este impuesto. 

La Ley Federal de Derechos (LFD) para el 2004 establece el cobro de este 
impuesto dependiendo del nivel de concentración de contaminantes del agua. 

Así, los tipos de cuerpo de agua se clasifican en tres, A, B y C, siendo los de 
tipo A los que permiten descargas con mayor concentración de contaminantes, 
los de tipo C, los que aceptan menos y B un nivel intermedio. Se calcula el 
volumen y la concentración de contaminantes descargados trimestralmente, de 
acuerdo a una regla determinada en la LFD, se revisan los resultados obtenidos 
de la medición y se comparan contra los límites permisibles publicados en la 
LFD, para cada tipo de contaminante. El monto a pagar se calcula en función de 
los niveles de contaminación que sobrepasan el límite permisible. La  LFD 
indica que “Los ingresos que se obtengan por la recaudación de estos derechos 
se destinarán a la Comisión Nacional del Agua para obras de infraestructura de 
saneamiento por cuenca hidrológica”. 

Cuando se piensa en corregir la ineficacia de la política de control de aguas 
residuales, es común pensar que incorporar una compensación al agente que 
recibe el daño mejoraría la efectividad del impuesto pigouviano, sin embargo la 
compensación por un lado inhibe los incentivos para protegerse de los efectos 
de la contaminación y por otro, induce a soluciones ineficientes. No es de 
extrañarse por tanto que la aplicación de la LFD no haya producido los efectos 
benéficos esperados. A continuación se evalúan los alcances de implantar un 
impuesto al uso consuntivo del agua. 
 

Modelo 

En la economía modelada en este trabajo se agrega el conjunto de bienes no 
relacionados directamente con el agua y se describe la parte hidráulica de la 
economía. Para ello se introduce un bien intermedio, agua útil, es decir el agua 
que entra realmente como insumo al proceso productivo, el cual es distinto al 
agua en bruto, de origen superficial o subterráneo, que se considera un factor 
de la producción. En todo proceso productivo se genera agua residual, misma 
que puede ser tratada o no. Si se trata puede servir como insumo para producir 
el bien de consumo, de lo contrario, afecta a la producción de agua útil, 
aumentando, la cantidad de valor añadido necesaria para producir agua útil. A 
continuación se describe el modelo de equilibrio general que se utilizará para 
evaluar los alcances del impuestos al uso consuntivo del agua.  

Bienes y factores 

Existen cuatro factores de producción: trabajo (L), capital (K),  agua superficial 
(As) y agua subterránea (At) y cinco bienes: bien producido con agua de primer 
uso (1); bien producido con agua tratada (2), agua útil de origen superficial (3), 
agua útil de origen subterráneo (4), agua tratada (5) y agua residual (6). Cabe 

mencionar que existe la posibilidad de no utilizar toda el agua, en este caso se 
supone que el agua se tira al mar o se pierde, no se puede ahorrar.  

Consumidores  

Hay dos tipos de consumidores: uno poseedor del agua en la cuenca, superficial 
y subterránea, además de una porción de trabajo y el otro, dueño del capital y 
de una porción de trabajo. Ambos consumidores maximizan su utilidad sujeta a 
su restricción presupuestaria, en un ambiente estático y de certidumbre. De la 
solución de estos problemas se obtienen las demandas marshallianas por los 
bienes 1 y 2, esto puede ser expresado de la siguiente manera: 
 

( ) 2,1;2,1,* , === jiIpfX ijji        (1) 
 
donde iI  es el ingreso del consumidor i, dado de la siguiente 

manera: RrKwLI 1111 θ++=  e RApAprKwLI tatsas 2222 θ++++= , X*i,j es la 
demanda del consumidor i por el bien j, pj es el precio del bien j, w es el precio 
del trabajo, r el precio del capital, pas es el precio del agua superficial,  pat el 
precio del agua subterránea, As la cantidad de agua superficial en la cuenca, At 
la cantidad de agua subterránea en la cuenca, θi es la proporción de la 
recaudación transferida al consumidor i; R, es la recaudación total del 
gobierno;  

Productores 

Los bienes 1, 2, 3 y 4 son producidos utilizando bienes y factores. Bajo el 
supuesto de que los productores minimizan los costos sujetos a una restricción 
tecnológica, se pueden obtener las demandas derivadas por factores y bienes, 
que se denominan de la siguiente manera: 
 

( )ijji
D YpfX ,,

* =          (2) 
 
y 
 

( )iataski
F YpprwfX ,,,,,

* =          (3) 
 
donde 
 
XD*

ij es la demanda derivada del bien j para producir el bien i; XF*
kj es la 

demanda derivada del factor k para producir el bien i; Yi cantidad producida 
del bien i. 

 Gobierno 

La función del gobierno, en esta economía, consiste sólo en recaudar impuestos 
y redistribuir la recaudación, a manera de transferencia lump-sum. Sólo existen 
dos fuentes de recaudación: los impuestos al trabajo (R1) y al consumo (R2). La 
recaudación total de la economía viene dada por la ecuación (4). 
 

∑∑∑ ++=+=
j

j
j

j
l

jlL XpXpXwRRR ,222,111,21 τττ       (4) 

Políticas impositivas a simular 

Se simulará la inclusión de dos tipos de impuestos: uno al uso consuntivo, es 
decir a la diferencia entre el agua útil y el agua residual; y otro al agua 
residual, en ambos casos se supondrá que la economía está previamente 
distorsionada con impuestos al trabajo (R1) y al consumo (R2). 

En el caso del impuesto al uso consuntivo, la incidencia impositiva se verá 
reflejada en un incremento al precio del agua útil por un monto igual a: 

( )53 YYtuc −  donde tuc es la tasa de impuesto ad valorem. 
En el segundo caso, es decir en el caso del impuesto al agua residual, tres, 

supondremos que el agua residual afecta la tecnología de producción de agua 
útil, incrementando la cantidad de recursos necesaria para producir la misma 
cantidad de agua útil. Esta externalidad se corrige gravando la producción de 
agua residual con una tasa ad valorem. 

Debido a la dificultad de medir con precisión el incremento de valor 
añadido necesario para compensar el daño causado con la introducción del agua 
residual en la producción de agua útil, se simularán distintas proporciones de 
daño. La forma funcional de la función de daño es la siguiente:  

 

( ) TY
YkXD

3

3
3 =           (5) 

 
Donde k es un factor de calibración y TY3 es la cantidad de agua residual a partir 
de la cual se produce el daño medioambiental. 

Equilibrio 

El equilibrio se describe en las siguientes ecuaciones, cabe mencionar que el 
gobierno salda sus cuentas de tal forma que el déficit es cero.  
 
Equilibrio en el mercado de bienes:    jYXX j

i

D
jiji

i

∀=+∑∑ ,,*    (6) 

 
Equilibrio en el mercado de factores:  
 

LX Li
F

i

=∑ ,
* ,         (7) 

KX Ki
F

i

=∑ ,
* ,         (8) 

AsX Asi
F

i

=∑ ,
* ,          (9) 

AtX Ati
F

i

=∑ ,
*          (10) 

 
y  
 
Déficit gubernamental cero: 
 

∑∑∑∑ ++=+=
= j

j
j

j
l

jlL
i

i XpXpXwRRR ,222,111,21

2

1

τττθ     (11) 

 

Especificación numérica 

Para que el modelo pueda ser utilizado, es necesario proponer una 
especificación numérica para cada una de las funciones, se usarán en este 
modelo formas funcionales Cobb-Douglas, para las funciones de utilidad y una 
especificación anidada Leontief - Cobb-Douglas para la parte de la producción. 
Lo cual puede expresarse de la siguiente manera: 
 

Demandas finales: 
j

iji
ji p

I
X ,*

,

α
=         (12) 

 
Con ji,α = la proporción del gasto del agente i en el bien j. 
 

Demandas intermedias: 












=
j

j

ij

jD
ji v

V
a
X

X ,,        (13) 

Donde  

=jia , el coeficiente técnico del insumo i para producir el bien j 

=jV  la cantidad de valor añadido para producir el bien j 

jv = el coeficiente técnico en la utilización del valor añadido. 

El valor de jV  viene dado por: 
 

4321 ββββ
TSjjjj AALKAV =          (14) 

Los parámetros de todas las ecuaciones anteriores deben ser determinados de 
tal modo que se alcancen las condiciones de equilibrio mediante la calibración 
del modelo. 

Calibración 

La calibración se hace tomando como referencia el equilibrio descrito en la 
base de datos y encontrando los parámetros que reproducen las condiciones de 
primer orden de los agentes, es decir, la información de las variables 

Tsjijijj
F
ji

D
jiji AALKVpXXX ,,,,,,,, ,,,,

*
,   

permite determinar el valor de: 

 21, ,,,, ττα jjijji Ava  y jβ .  

En términos llanos, se trata de encontrar los parámetros que reproduzcan 
las condiciones de primer orden de los agentes. Una vez hecho esto, es posible 
calcular nuevos equilibrios con las condiciones que se quieren simular. La 
determinación de los nuevos equilibrios se hará con el algoritmo descrito en el 
siguiente inciso.   

Solución 

El modelo se resuelve utilizando el método iterativo de Shoven y Whalley 
(1984): 1) se conjeturan precios por los factores, con lo cual es posible calcular 
el valor de las dotaciones de factores de cada uno de los agentes, se conjetura 
también la recaudación esperada; 2) se calculan las demandas de bienes finales 
de cada uno de los agentes, una vez conocido su ingreso; 3) la suma de 
demandas finales debe ser igual a la producción de bienes; 4) se calculan las 
demandas derivadas de factores necesarias para producir la cantidad de bienes 
ya calculados. Si son iguales a las dotaciones de la economía, entonces se ha 
encontrado un equilibrio, de lo contrario se modifican los precios de factores y 
se vuelve a iterar hasta alcanzar el equilibrio. 

En el anexo B se reporta uno de los archivos programados en GAMS en donde 
se calibran, resuelven y simulan las distintas posibilidades de impuestos. 

Resultados 

El modelo fue calibrado para dos casos, la cuenca Grijalva-Usuamacinta y la 
Nazas-Aguanaval, los resultados obtenidos de la calibración han de compararse 
con los resultados simulados y, de esta manera, evaluar la bondad de las 
políticas en términos económicos y en términos hidrológicos. 

La incidencia de los dos impuestos evaluados se analizará desde dos puntos 
de vista: a) se medirá el efecto de los impuestos sobre la cantidad de agua 
reciclada y la emisión de agua residual, efectos hidrológicos y b) en relación a 
la consecuencia de la aplicación de los impuestos sobre el bienestar social, 
efectos económicos.  

Efectos hidrológicos 

La cuenca Grijalva-Usumacinta es una cuenca con gran abundancia de agua y 
poco desarrollo industrial, mientras que la cuenca Nazas-Aguanaval es más 
industrializada y presenta grandes problemas de escasez relativa de agua.  

En las tablas 1 y 2, se muestran los resultados obtenidos por la aplicación de 
los impuestos en el caso Grijalva-Usumacinta sobre los niveles de equilibrio en 
agua residual, agua tratada, agua útil de origen superficial y agua útil de origen 
subterráneo. 

 
T A B L A  1  

GRIJALVA-USUMACINTA 

INCREMENTO PORCENTUAL EN LA CANTIDAD DE AGUA RESIDUAL Y AGUA TRATADA PRODUCIDA COMO 

CONSECUENCIA DEL INCREMENTO PORCENTUAL EN EL IMPUESTO AL USO CONSUNTIVO, TUC 

tuc
A gua 

residual
A gua 
tratada

0.1 0.44 0.45
0.2 0.89 0.89
0.3 1.33 1.34
0.4 1.78 1.79
0.5 2.14 2.16
0.6 2.58 2.53
0.7 2.94 2.90
0.8 3.29 3.27
0.9 3.56 3.65
1 3.91 3.94  

T A B L A  2  

GRIJALVA-USUMACINTA 
INCREMENTO PORCENTUAL EN LA CANTIDAD DE AGUA RESIDUAL Y AGUA TRATADA PRODUCIDA COMO 

CONSECUENCIA DEL INCREMENTO PORCENTUAL EN EL IMPUESTO AL AGUA RESIDUAL, TRES 

tres
Agua 

residual
Agua 

tratada
0.01 -0.71 -0.74
0.02 -1.51 -1.49
0.03 -2.22 -2.23
0.04 -3.11 -3.13
0.05 -4.00 -4.02
0.06 -4.98 -4.99
0.07 -6.05 -6.03
0.08 -7.21 -7.22
0.09 -8.45 -8.48
0.10 -9.88 -9.90  

 
 

Como puede observarse de la información de las tablas anteriores, en la 
cuenca Grijalva-Usumacinta, donde hay una gran abundancia de agua, cuando 
se grava el uso consuntivo, es decir la diferencia entre agua útil y agua 
residual, el resultado obtenido es un incremento en la producción de agua 
residual y consecuentemente una mayor utilización de agua tratada. Sin 
embargo, cuando se aplica el impuesto al agua residual, tres disminuye la 
producción de ésta y por ende la del agua tratada. En el caso de la cuenca 
Grijalva-Usumacinta, no es posible disminuir la producción de agua residual y al 
mismo tiempo incrementar la reutilización de agua tratada en ninguno de los 
instrumentos de política propuestos. 

En el caso de la cuenca Nazas-Aguanaval, cuenca con gran escasez de agua, 
los resultados obtenidos son más optimistas: 

T A B L A  3  

NAZAS-AGUANAVAL 
INCREMENTO PORCENTUAL EN LA CANTIDAD DE AGUA RESIDUAL Y AGUA TRATADA PRODUCIDA COMO 

CONSECUENCIA DEL INCREMENTO PORCENTUAL EN EL IMPUESTO AL USO CONSUNTIVO, TUC 

tuc
Agua 

residual
Agua 

tratada
0.1 -0.39 -0.40
0.2 -0.73 -0.76
0.3 -1.07 -1.04
0.4 -1.31 -1.28
0.5 -1.50 -1.53
0.6 -1.70 -1.71
0.7 -1.89 -1.89
0.8 -2.03 -2.02
0.9 -2.18 -2.17

1 -2.33 -2.32  
 
 
 

T A B L A  4  

NAZAS-AGUANAVAL 
INCREMENTO PORCENTUAL EN LA CANTIDAD DE AGUA RESIDUAL Y AGUA TRATADA PRODUCIDA COMO 

CONSECUENCIA DEL INCREMENTO PORCENTUAL EN EL IMPUESTO AL USO CONSUNTIVO, TUC 

tres

Agua 
residual

Agua 
tratada

0.10 -1.60 -1.62
0.20 -3.34 -3.33
0.30 -5.43 -5.44
0.40 -8.62 -8.62
0.50 -14.78 -14.79
0.60 -34.25 -34.26  

 
Si bien el efecto sobre la producción de agua residual es el mismo al 

observado en el caso de Grijalva-Usumacinta, es decir, como es de esperarse, 
al incrementarse el gravamen por producir esta externalidad, se inhibe su 
producción, el efecto del impuesto al uso consuntivo se invierte en relación a 
los resultados obtenidos en la simulación para la cuenca Grijalva-Usumacinta.  

Lo anterior puede explicarse debido a los precios relativos. Dada la gran 
escasez de agua, al gravar el uso consuntivo, se incrementa el precio del agua 
útil y por tanto se disminuye su demanda, consecuentemente disminuye la 
cantidad de agua intercambiada en la economía. En este caso, tanto gravar el 
uso consuntivo como gravar el agua residual, producen el mismo efecto en la 
emisión de agua residual. Sin embargo, como el impuesto al agua residual, 

grava directamente a este bien, su efecto es mucho mayor que el efecto 
producido por el impuesto al uso consuntivo. 

Efectos económicos 

Para analizar los efectos económicos de la aplicación de estas políticas 
públicas, se calculó la variación equivalente sobre cada uno de los agentes que 
forman la economía y posteriormente se sumaron para calcular la variación 
equivalente total. Aprovechando que las funciones de utilidad de los agentes 
son homotéticas, la variación equivalente puede calcularse de la siguiente 
manera, Shoven y Whalley(1984):  
 

0
0

01 I
U
UU

VEi 






 −
=          (15) 

 
y por tanto la variación equivalente para la sociedad es:  
 

∑=
i

iVEVET           (16) 

 
Donde U1 es el nivel de utilidad alcanzado después de la aplicación de los 

impuestos, U0 el nivel de utilidad inicial e I0 el ingreso inicial. 
La variación de las tasas de impuesto son distintas en magnitud cuando se 

aplican al uso consuntivo que al agua residual porque, en este caso, el efecto 
sobre los precios es indirecto, mientras que cuando se aplica sobre la 
producción de agua residual, el efecto sobre los precios es directo y por tanto 
los precios relativos de los bienes se ven más afectados. Así, las tasas 
impositivas para el caso del uso consuntivo pueden variar entre el 10 y 100%, 
mientras que en el caso del agua residual, solo pueden variar en el rango de 1 
al 10%.  

Para el caso de la cuenca Grijalva-Usumacinta, los resultados se modifican 
en cuanto a la aplicación del impuesto al uso consuntivo, éstos pueden 
observarse en los cuadros 5 y 6.  

La compensación equivalente a nivel social, se incrementa conforme 
aumenta la tasa de gravamen, sin embargo, la compensación para el agente H1 
es positiva, mientras que para el agente H2, poseedor del agua, es negativa. En 
ningún momento la compensación positiva alcanza a modificar el signo de la 
compensación total. 

Por otro lado, en el caso del impuesto al agua residual, el comportamiento 
de la economía es similar en ambos casos, el agente que no posee agua ve 
disminuido su bienestar, lo cual puede ser visto como la necesidad de cada vez 
mayores compensaciones, mientras que el agente poseedor del agua se ve 

beneficiado de la aplicación de un impuesto al agua residual, este beneficio 
tiene un máximo alrededor del 5% y una vez sobrepasado éste, empieza a 
decaer el bienestar social, como consecuencia de una reducción excesiva de la 
producción de bienes asociados a la producción de agua residual.  

 
T A B L A  5  

GRIJALVA-USUMACINTA 
VALORES DE LA VARIACIÓN EQUIVALENTE COMO CONSECUENCIA DEL INCREMENTO PORCENTUAL 

EN EL IMPUESTO AL USO CONSUNTIVO, TCON 

tuc H1 H2 Total
0.1 1.561 -2.118 -0.557
0.2 3.038 -4.164 -1.126
0.3 4.431 -6.131 -1.7
0.4 5.741 -8.007 -2.266
0.5 6.965 -9.784 -2.819
0.6 8.103 -11.452 -3.349
0.7 9.155 -13.003 -3.848
0.8 10.119 -14.432 -4.313
0.9 10.998 -15.734 -4.736

1 11.792 -16.908 -5.116  

 
 

T A B L A  6  

GRIJALVA-USUMACINTA 
VALORES DE LA VARIACIÓN EQUIVALENTE COMO CONSECUENCIA DEL INCREMENTO PORCENTUAL 

EN EL IMPUESTO AL USO CONSUNTIVO, TRES 

tres H1 H2 Total
0.01 -2.743 4.611 1.868
0.02 -5.516 8.798 3.282
0.03 -8.326 12.59 4.264
0.04 -11.182 16.008 4.826
0.05 -14.095 19.068 4.973
0.06 -17.077 21.775 4.698
0.07 -20.145 24.128 3.983
0.08 -23.32 26.116 2.796
0.09 -26.63 27.717 1.087
0.1 -30.111 28.892 -1.219  

T A B L A  7  

NAZAS-AGUANAVAL 
VALORES DE LA VARIACIÓN EQUIVALENTE COMO CONSECUENCIA DEL INCREMENTO PORCENTUAL 

EN EL IMPUESTO AL USO CONSUNTIVO, TCON 

tuc H 1 H 2 Total

0.1 -0.626 -1.496 -2.122
0.2 -1.146 -2.771 -3.917
0.3 -1.588 -3.874 -5.462
0.4 -1.968 -4.839 -6.807
0.5 -2.301 -5.692 -7.993
0.6 -2.593 -6.452 -9.045
0.7 -2.854 -7.133 -9.987
0.8 -3.087 -7.748 -10.835
0.9 -3.298 -8.306 -11.604
1 -3.488 -8.815 -12.303  

 
 
 

T A B L A  8  

NAZAS-AGUANAVAL 
VALORES DE LA VARIACIÓN EQUIVALENTE COMO CONSECUENCIA DEL INCREMENTO PORCENTUAL 

EN EL IMPUESTO AL USO CONSUNTIVO, TRES 

 

tre s H 1 H 2 Total

0.01 -5.162 5.861 0.699
0.02 -10.298 11.349 1.051
0.03 -15.409 16.5 1.091
0.04 -20.496 21.348 0.852
0.05 -25.559 25.918 0.359
0.06 -30.6 30.235 -0.365
0.07 -35.618 34.32 -1.298

N azas-A guanaval. V ariación equivalente en relación 
al equilibrio de referencia

 
 
 

En las tablas 7 y 8, se muestran los resultados en términos de variación 
equivalente. Nótese que cuando se aplica el impuesto al uso consuntivo, se 
aumenta la compensación que se tendría que dar a los agentes para dejarlos en 
la misma situación de bienestar que se encontraban antes de la aplicación de 
las políticas. Debe recordarse, ver la tabla A1 y A2, del anexo A, que el agente 
poseedor del agua, es el agente H2. Cuando se aplica el impuesto al uso 
consuntivo, y por tanto se encarece el agua útil, el poseedor del agua ve 
disminuido su bienestar, mientras que cuando se grava la producción de agua 

residual, se restauran las condiciones de eficiencia social, compensando al 
agente poseedor del agua hasta el nivel óptimo, una vez sobrepasado este 
nivel, el costo de seguir gravando supera el beneficio derivado de la 
compensación y el bienestar social cae. 

Conclusiones 

Se construyó un modelo de equilibrio general computable con el fin de evaluar 
distintas políticas públicas: un impuesto al uso consuntivo y un impuesto al agua 
residual, en ambos casos se siguió la metodología propuesta por Shoven y 
Whalley. Se calibraron dos economías con dotaciones de agua muy distintas. 

Al simular el comportamiento de las economías previamente distorsionadas 
con impuestos al trabajo y al consumo, se llegó a las siguientes conclusiones: 

 
• Los efectos hidrológicos y económicos derivados del impuesto al agua 

residual y el impuesto al uso consuntivo son instrumentos de política con 
resultados, en general, distintos.  

• El impuesto al agua residual afecta mucho más directamente a los 
precios y por ende, sus efectos son más dramáticos.  

• El efecto del impuesto al uso consuntivo es indirecto y por tanto las tasas 
a las que puede aplicarse son mayores.  

• Con fines de disminuir la emisión de agua residual, el impuesto directo a 
este “mal”, es más conveniente.  

• El impuesto al uso consuntivo no resultó muy eficiente como forma de 
incentivar el reciclamiento de agua.  

• En términos de bienestar social el impuesto al agua residual puede 
resultar mejor que el impuesto al uso consuntivo, si se aplica en la 
medida correcta. 

ANEXO A 

En este anexo se muestran las dos matrices utilizadas para la calibración del 
modelo, estas matrices intentan reflejar en la medida de lo posible los 
equilibrios que se dan en las cuencas Grijalva-Usumacinta y Nazas-Aguanaval.  

Las matrices fueron construidas con información del Instituto Nacional de 
Estadística Geografía e Informática, del Instituto Mexicano de Tecnología del 
Agua y la Comisión Nacional del Agua, no son el resultado de una aplicación 
sistemática de las metodologías publicadas para construir una matriz de 
contabilidad social, no podría serlo, debido al hecho de que no existe 
información económica a nivel de cuenca. 
 
 

C U A D R O  A 1  

DATOS DEL EQUILIBRIO DE REFERENCIA 
ECONOMÍA SIMPLIFICADA DE LA CUENCA GRIJALVA-USUMACINTA 

1. B ien seco 269.1 550.8 276.6 108.9

2. B ien húm edo 500.7 2333.7 246.0 183.0

3. A gua residual 112.4

4. A gua tratada 134.4

5. A gua útil 

superficial 10.2 19.3 60.0

6. A gua útil 

subterránea 10.0 14.2 27.0

7. A gua 

superficial no 

usada 355.5

8. A gua 

subterránea no 

usada 9.9

9. Trabajo 180.5 94.5 15.4 15.0 27.2 29.1 8.7 0.2

10. C apital 221.7 107.5 10.0 7.0 28.0 19.2 19.0 0.2

11. A gua 

superficial 28.5 327.0

12. A gua 

subterránea 0.4 9.5

13. A gente 1 100.0 412.6 10.0

14. A gente 2 270.6 355.5 9.9 21.3

15. Im puesto al 

trabajo 11.2 6.0 5.8 2.5 0.8

16. Im puesto al 

consum o 2.0 3.0
17. R ecaudación 

total 26.3 5.0

C U A D R O  A 2  

DATOS DEL EQUILIBRIO DE REFERENCIA 
ECONOMÍA SIMPLIFICADA DE LA CUENCA NAZAS-AGUANAVAL 

1 2 3 4 5 6 8 9 10 11 12 13 14 15 16 17

1. B ien seco 551.6 1187.0 751.3 396.6

2. B ien húm edo 1459.4 1408.8 1529.2 370.6

3. A gua residual 206.4

4. A gua tratada 327.2

5. A gua útil 

superficial 12.1 20.0 101.0

6. A gua útil 

subterránea 7.4 15.0 86.6

8. A gua 

subterránea no 

usada 49.1

9. Trabajo 230.2 542.0 10.0 50.0 45.7 54.0 11.0

10. C apital 597.4 1208.4 8.0 70.0 43.0 50.0 30.8

11. A gua 

superficial 37.6

12. A gua 

subterránea 4.2 7.3

13. A gente 1 242.9 2007.6 30.0

14. A gente 2 700.0 37.6 11.5 67.2

15. Im puesto al 

trabajo 18.4 49.6 0.8 0.8 6.8 0.8

16. Im puesto al 

consum o 10.0 10.0
17. R ecaudación 

total 77.2 20.0

ANEXO B 

$TITLE  Modelo de AGUA USO CONSUNTIVO GRIJALVA USUMACINTA 
$ONDIGIT 
$OFFSYMXREF OFFSYMLIST OFFUELLIST OFFUELXREF 
* 
* 
* 01.09.04 
* Modelo para simular el impuesto al uso consuntivo y el impuesto 
* al agua residual 
* 
 OPTIONS LIMCOL=0, LIMROW=0; 
SETS 
   TOTAL total de indices usados / S1*S8, AP,AT,L,K, H1*H2, R1*R3, TOT / 
   S(TOTAL)  bienes               / S1*S8 / 
   LK(TOTAL) factores             / L,K,AP,AT / 
   H(TOTAL)  consumidores         / H1*H2  / 
   R(TOTAL)  impuestos            / R1*R3  / 
   SX(S)                             /S5/ 
   SY(S)                             /S1,S2,S3,S4,S6,S7,S8/ 
   ; 
   ALIAS (TOTAL,TOTA); 
   ALIAS (S,SS); 
   ALIAS (LK,KL); 
 
TABLE SAM(TOTAL,TOTA) Matriz de Contabilidad Social 
          S1     S2    S3    S4     S5    S6   S7   S8     L     K    AP  AT 
   S1  269.1  550.8 
   S2  500.7 2333.7 
   S3                     112.4 
   S4         134.4 
   S5 
   S6 
   S7   10.2   19.3  60.0 
   S8   10.0   14.2  27.0 
   L   180.5   94.5  15.4  15.0   8.7    0.2 27.2 29.1 
   K   221.7  107.5  10.0   7.0  19.0    0.2 28.0 19.2 
   AP                           327.0        28.5 
   AT                                    9.5       0.4 
   H1                                                  100.0 412.6 
   H2                                                  270.6       355.5 9.9 
   R1   11.2    6.0   .00    .0   0.8    0.0  5.8  2.5 
   R2    2.0    3.0 
   R3 
 
  +      H1    H2        R1   R2  R3 
   S1 276.6 108.9 
   S2 246.0 183.0 
   S3 
   S4 
   S5       355.5 
   S6        09.9 
   S7 
   S8 
   L 
   K 
   AP 
   AT 
   H1                              10.0 
   H2                              21.3 
   R1 
   R2 
   R3                  26.3  5.0 

    ; 
 
PARAMETERS 
   TETA(H) 
   LEON(S,SS) 
   NUVALEON(S) 
   VALEON(S) 
   BETA(LK,S) 
   A(S) 
   ALFA(S,H) 
   TL(LK,S) 
   TC(S) 
   TT(H) 
   INC0(H) 
   U0(H) 
   INC1(H) 
   U1(H) 
 
   ; 
* * * * * * * * * ETAPA 1 CALIBRACION * * * * * * * * * * 
*CALIBRACION DE LAS DOTACIONES 
 
   SAM('TOT',TOTAL) = SUM(TOTA, SAM(TOTA,TOTAL)); 
   SAM(TOTAL,'TOT') = SUM(TOTA, SAM(TOTAL,TOTA)); 
   DISPLAY SAM; 
 
*CALIBRACION DE LOS PARAMETROS DE LA TECNOLOGIA 
   LEON(S,SS) = 0.0; 
   LEON(S,SS) = SAM(S,SS)/(SAM('TOT',SS)-SAM('R2',SS))$ 
                (SAM('TOT',SS)-SAM('R2',SS) NE 0); 
   DISPLAY LEON; 
   NUVALEON(S) = ( SUM(LK,SAM(LK,S))+SAM('R1',S)) ; 
   VALEON(S)   = (SUM(LK,SAM(LK,S))+SAM('R1',S))/ 
                 (SAM('TOT',S)-SAM('R2',S))$ 
                 ((SAM('TOT',S)-SAM('R2',S)) NE 0); 
   DISPLAY NUVALEON; 
   DISPLAY VALEON; 
 
*CALIBRACION PARAMETROS FISCALES: IMPUESTO AL TRABAJO 
    TL(LK,S)=0.0; 
    TL('L',S)=SAM('R1',S)/SAM('L',S); 
    DISPLAY TL; 
 
* DESPEJE DE BETA 
BETA(LK,S)=(1+TL(LK,S))*SAM(LK,S)/SUM(KL,(1+TL(KL,S))*SAM(KL,S)); 
A(S)=NUVALEON(S)/PROD(LK,SAM(LK,S)**BETA(LK,S)); 
 
*CALIBRACIÓN PARÁMETROS DE CONSUMO 
ALFA(S,H) = SAM(S,H)/SAM('TOT',H)$( SAM('TOT',H) NE 0); 
 
*CALIBRACION PARAMETROS FISCALES: IMPUESTO AL CONSUMO 
    TC(S)$(SUM(H,SAM(S,H)) NE 0) = 
                 SAM('R2',S)/SUM(H,SAM(S,H)); 
     DISPLAY TC; 
*CALIBRACION PARAMETROS FISCALES: TRANSFERENCIAS 
    TT(H)=SAM(H,'R3')/SAM('R3','TOT'); 
     DISPLAY TT; 
DISPLAY BETA; 
DISPLAY A; 
DISPLAY ALFA; 
 
 
* * * * * * * * * ETAPA 2 VALIDACION BENCHMARK EQUILIBRIUM * * * * * * * * * 
 
   POSITIVE VARIABLES 

   XSH(S,H) 
   XSS(S,SS) 
   XLS(LK,S) 
   Y(S) 
   VA(S) 
   PS(S) 
   PL(LK) 
   ; 
   VARIABLES 
   RE 
   FO2 
   ; 
   EQUATIONS 
   DEMFIN(S,H)   ecuaciones para demanda final 
   DEMINT(S,SS)  ecuaciones para demanda intermedia 
   DEMAVA(S)     demanda de valor anyadido 
   DEMFAC1(LK,S) ecuaciones de factores 
   EQUIBIE(S)    condiciones de equilibrio para bienes 
   EQUIFAC(LK)   condiciones de equilibrio para factores 
   PRECIOS(S)    ecuaciones de precios 
   PRECIOSL(LK) 
   PREFIJO       precio fijo P(L) 
   RECAUDA       recaudacion 
   FUNOBJ2       funcion objetivo de la ETAPA 2 
   ; 
 
*DEMANDA FINAL 
*   DEMFIN(S,H) .. XSH(S,H)*PS(S)*(1+TC(S)) =E= 
*                  ALFA(S,H)*( SUM(LK, PL(LK)*SAM(H,LK))+TT(H)*RE); 
 
   DEMFIN(S,H) .. XSH(S,H)*PS(S) =E= 
                  ALFA(S,H)*( SUM(LK, PL(LK)*SAM(H,LK))+TT(H)*RE); 
 
*DEMANDA INTERMEDIA 
   DEMINT(S,SS) ..XSS(S,SS) =E= LEON(S,SS)*(Y(SS)-SAM('R2',SS)); 
 
*DEMANDA VALOR ANYADIDO 
   DEMAVA(S) ..VA(S) =E= VALEON(S)*(Y(S)-SAM('R2',S)); 
 
*DEMANDA DE FACTORES 
DEMFAC1(LK,S) ..(1/A(S))*(BETA(LK,S)/(PL(LK)*(1+TL(LK,S))))*VA(S) =E= 
             XLS(LK,S)*PROD(KL,(BETA(KL,S)/(PL(KL)*(1+TL(KL,S))))**BETA(KL,S)); 
 
*CONDICIONES DE EQUILIBRIO PARA BIENES 
EQUIBIE(S)  ..Y(S) =E= SUM(H, XSH(S,H)) + SUM(SS, XSS(S,SS)); 
 
*EQUIBIE(S)  ..Y(S) + SUM(H, XSH(S,H))*(TC(S)/(1+TC(S))) =E= 
*                     SUM(H, XSH(S,H)) + SUM(SS, XSS(S,SS)); 
 
*CONDICIONES DE EQUILIBRIO PARA FACTORES 
EQUIFAC(LK) .. SUM(S, XLS(LK,S)) - SAM('TOT',LK) =E= 0 ; 
 
* PRECIOS 
PRECIOS(S) ..PS(S)*(Y(S)-SAM('R2',S)) =E= SUM( SS, PS(SS)*XSS(SS,S) ) + 
                        SUM ( LK, PL(LK)*(1+TL(LK,S))*XLS(LK,S) ) ; 
 
*RECAUDACION 
RECAUDA..  RE =E=PL('L')*SUM(S,TL('L',S)*XLS('L',S))+ 
                  SUM(H,SUM(S,TC(S)*PS(S)*XSH(S,H))); 
 
*PRECIO FIJO P(L) 
PREFIJO ..PL('L') =E= 1. ; 
 
*FUNCION OBJETIVO 
FUNOBJ2 ..FO2 =E= SUM( S, PS(S)*( SUM(H,XSH(S,H)) ) )+ 

               SUM( LK,PL(LK)*( SUM (S,XLS(LK,S) ) )); 
 
 
* VALORES INICIALES DE LAS VARIABLES 
   PS.L(S)= 1.0; 
   PL.L(LK)=1.0; 
   XSH.L(S,H)=SAM(S,H); 
   XSS.L(S,SS)=SAM(S,SS); 
   XLS.L(LK,S)=SAM(LK,S); 
   Y.L(S)= SAM('TOT',S); 
   VA.L(S)=NUVALEON(S); 
 
MODEL BENCHMARK /DEMFIN,DEMINT,DEMAVA,DEMFAC1,EQUIBIE,EQUIFAC, 
                 PRECIOS,PREFIJO,RECAUDA,FUNOBJ2 / 
OPTION ITERLIM = 1000; 
SOLVE BENCHMARK MINIMIZING FO2 USING NLP; 
    INC0(H)= SUM(LK, PL.L(LK)*SAM(H,LK))+TT(H)*RE.L; 
    U0(H)= PROD(S,XSH.L(S,H)**ALFA(S,H)); 
 
 
* * * * * * * * ETAPA 3 SIMULACION COUNTERFACTUAL EQUILIBRIUM * * * * * * * * 
* * * * * * * * Impuesto al uso consuntivo con danyo medioambiental  * * * * 
 
PARAMETERS 
   VAFI(S) 
   TAXUC(S) 
   TAXRES(S) 
   DIST(H) 
   INCOME(H) 
   YANT 
   VEQ(H) 
   VEQT 
   GAMA 
   ; 
   VAFI(S)=VA.L(S); 
   TAXUC(S)=0.0; 
   TAXUC('S5')=.0; 
   TAXRES(S)=0.; 
   TAXRES('S3')=0.0; 
   DIST('H1')=0.571; 
   DIST('H2')=0.429; 
   YANT=Y.L('S3'); 
   GAMA=1.0; 
   POSITIVE VARIABLES 
   REC 
   ; 
   VARIABLES 
   FO3 
   ; 
   EQUATIONS 
   DEMINT3(S,SS)  demanda intermedia de bienes 
   DEMAVA3(S)     demanda de valor anyadido 
   DEMFAC3(LK,S)  ecuaciones de factores 
   EQUIBIE2(S)    condiciones de equilibrio para bienes 
   EQUIFAC2(LK) 
   PRECIOS2(S)    ecuaciones de precios 
   PREFIJO3       precio fijo 
   RECAUDA2       recaudacion 
   FUNOBJ3        funcion objetivo de la ETAPA 3 
   DEMFIN2(S,H)  DEMANDA FINAL 
   ; 
 
*DEMANDA FINAL 
DEMFIN2(S,H) .. XSH(S,H)*PS(S) =E= 
                  ALFA(S,H)*( SUM(LK, PL(LK)*SAM(H,LK))+TT(H)*RE); 

*DEMANDA INTERMEDIA 
   DEMINT3(S,SS) ..XSS(S,SS) =E= LEON(S,SS)*(Y(SS)-SAM('R2',SS)); 
 
*LA PRODUCCION DE AGUA RESIDUAL INCREMENTA LA DEMANDA DE FACTORES 
 
*DEMANDA VALOR ANYADIDO 
DEMAVA3(S) ..VA(S) =E= VALEON(S)*(Y(S)-SAM('R2',S))$SY(S)+ 
                      GAMA*(Y('S3')/YANT)*VALEON(S)*(Y(S)-SAM('R2',S))$SX(S); 
 
*DEMANDA DE FACTORES 
DEMFAC3(LK,S) ..(1/A(S))*(BETA(LK,S)/(PL(LK)*(1+TL(LK,S))))*VA(S) =E= 
             XLS(LK,S)*PROD(KL,(BETA(KL,S)/(PL(KL)*(1+TL(KL,S))))**BETA(KL,S)); 
 
*CONDICIONES DE EQUILIBRIO PARA BIENES 
EQUIBIE2(S)  ..Y(S) =E= SUM(H, XSH(S,H)) + SUM(SS, XSS(S,SS)); 
 
 
*CONDICIONES DE EQUILIBRIO PARA FACTORES 
EQUIFAC2(LK) .. SUM(S, XLS(LK,S)) - SAM('TOT',LK) =E= 0 ; 
 
* PRECIOS 
PRECIOS2(S) ..PS(S)*(Y(S)-SAM('R2',S)) =E= 
                        (1+TAXRES('S3'))*(SUM( SS, PS(SS)*XSS(SS,S) ) + 
                        SUM ( LK, PL(LK)*(1+TL(LK,S))*XLS(LK,S) ) + 
                        TAXUC(S)*(Y('S5')-Y('S3'))); 
 
*PRECIO FIJO P(L) 
PREFIJO3 ..PL('L') =E= 1. ; 
 
*RECAUDACION 
RECAUDA2..  REC =E=PL('L')*SUM(S,TL('L',S)*XLS('L',S))+ 
                  SUM(H,SUM(S,TC(S)*PS(S)*XSH(S,H)))+ 
                  PS('S5')*TAXUC('S5')*(Y('S5')-Y('s3'))+ 
                  PS('S3')*TAXRES('S3')*Y('S3'); 
 
*FUNCION OBJETIVO 
FUNOBJ3 ..FO3 =E= SUM( S, PS(S)*( SUM(H,XSH(S,H)) ) )+ 
               SUM( LK,PL(LK)*( SUM (S,XLS(LK,S) ) )); 
 
* VALORES INICIALES DE LAS VARIABLES 
   PS.L(S)= 1.0; 
   PL.L(LK)=1.0; 
   XSH.L(S,H)=SAM(S,H); 
   XSS.L(S,SS)=SAM(S,SS); 
   XLS.L(LK,S)=SAM(LK,S); 
   Y.L(S)= SAM('TOT',S); 
VA.L(S)=VAFI(S); 
MODEL COUNTERFAC /DEMFIN2,DEMFAC3,DEMINT3,DEMAVA3, 
                 EQUIBIE2,EQUIFAC2, 
                 PRECIOS2,RECAUDA2,PREFIJO3,FUNOBJ3 / 
 
OPTION ITERLIM = 1000; 
SOLVE COUNTERFAC MINIMIZING FO3 USING NLP; 
 
  INC1(H)= SUM(LK, PL.L(LK)*SAM(H,LK))+TT(H)*RE.L; 
  U1(H)= PROD(S,XSH.L(S,H)**ALFA(S,H)); 
 
 
DISPLAY INC0; 
DISPLAY INC1; 
DISPLAY U0; 
DISPLAY U1; 
 
  VEQ(H)=(U1(H)-U0(H))*INC0(H)/U0(H); 

  VEQT = SUM(H,VEQ(H)); 
 
DISPLAY VEQ; 
DISPLAY VEQT; 

